Homogene lineare Differentialgleichungen

■ Übungsaufgabe 1

Wir betrachten die Differentialgleichung

In[1]:=
```
DE = y''''[x] + y[x] == 0
```

Out[1]= $y^{(4)}(x) + y(x) = 0$

mit dem charakteristischen Polynom

In[2]:=
```
charpol = DE /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```

Out[2]= $\lambda^4 + 1 = 0$

Das charakteristische Polynom hat die komplexen Lösungen:

In[3]:=
```
lösung = Solve[charpol]
```

Out[3]= $\left\{\left\{\lambda \to -\sqrt[4]{-1}\right\}, \left\{\lambda \to \sqrt[4]{-1}\right\}, \left\{\lambda \to -(-1)^{3/4}\right\}, \left\{\lambda \to (-1)^{3/4}\right\}\right\}$

In[4]:=
```
lösung = MapAll[ComplexExpand, lösung]
```

Out[4]= $\left\{\left\{\lambda \to -\frac{1+i}{\sqrt{2}}\right\}, \left\{\lambda \to \frac{1+i}{\sqrt{2}}\right\}, \left\{\lambda \to \frac{1-i}{\sqrt{2}}\right\}, \left\{\lambda \to -\frac{1-i}{\sqrt{2}}\right\}\right\}$

Entsprechend erhalten wir die Lösungsbasis der Differentialgleichung DE:

In[5]:=
```
komplexebasis = Map[Exp[# x] &, λ /. lösung]
```

Out[5]= $\left\{e^{-\frac{(1+i)x}{\sqrt{2}}}, e^{\frac{(1+i)x}{\sqrt{2}}}, e^{\frac{(1-i)x}{\sqrt{2}}}, e^{-\frac{(1-i)x}{\sqrt{2}}}\right\}$

In[6]:=
```
reellebasis = Union[ComplexExpand[Re[komplexebasis]], ComplexExpand[Im[komplexebasis]]]
```

Out[6]= $\left\{e^{-\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right), e^{\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right), -e^{-\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right), e^{-\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right), -e^{\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right), e^{\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right)\right\}$

In[7]:=
```
reellebasis = {reellebasis[[1]], reellebasis[[2]], reellebasis[[4]], reellebasis[[6]]}
```

Out[7]= $\left\{e^{-\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right), e^{\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right), e^{-\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right), e^{\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right)\right\}$

DSolve liefert:

In[8]:=
```
DSolve[DE, y[x], x]
```

Out[8]= $\left\{\left\{y(x) \to c_3\, e^{-\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right) + c_4\, e^{\frac{x}{\sqrt{2}}} \sin\left(\frac{x}{\sqrt{2}}\right) + c_1\, e^{\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right) + c_2\, e^{-\frac{x}{\sqrt{2}}} \cos\left(\frac{x}{\sqrt{2}}\right)\right\}\right\}$

Wir lösen nun ein Anfangswertproblem mit y(0)=0, y'(0)=1, y''(0)=0, y'''(0)=1:

In[9]:=
```
WronskiMatrix[basis_, x_] :=
 Table[D[basis[[k]], {x, j}], {j, 0, Length[basis] - 1}, {k, Length[basis]}]

WronskiDeterminante[basis_, x_] := Det[WronskiMatrix[basis, x]]
```

In[11]:=
```
WronskiMatrix[reellebasis, x] /. x → 0
```

Out[11]= $\begin{pmatrix} 1 & 1 & 0 & 0 \\ -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \\ 0 & 0 & -1 & 1 \\ \frac{1}{\sqrt{2}} & -\frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} & \frac{1}{\sqrt{2}} \end{pmatrix}$

In[12]:=
```
gleichungssystem = (WronskiMatrix[reellebasis, x] /. x → 0).
   {{c[1]}, {c[2]}, {c[3]}, {c[4]}} == {{0}, {1}, {0}, {1}}
```

Out[12]=
$$\begin{pmatrix} c(1)+c(2) \\ -\frac{c(1)}{\sqrt{2}}+\frac{c(2)}{\sqrt{2}}+\frac{c(3)}{\sqrt{2}}+\frac{c(4)}{\sqrt{2}} \\ c(4)-c(3) \\ \frac{c(1)}{\sqrt{2}}-\frac{c(2)}{\sqrt{2}}+\frac{c(3)}{\sqrt{2}}+\frac{c(4)}{\sqrt{2}} \end{pmatrix} = \begin{pmatrix} 0 \\ 1 \\ 0 \\ 1 \end{pmatrix}$$

In[13]:=
```
LinearSolve[(WronskiMatrix[reellebasis, x] /. x → 0), {0, 1, 0, 1}]
```

Out[13]= $\left\{0,\ 0,\ \frac{1}{\sqrt{2}},\ \frac{1}{\sqrt{2}}\right\}$

In[14]:=
```
sol = Solve[Flatten[(WronskiMatrix[reellebasis, x] /. x → 0).{{a}, {b}, {c}, {d}}] ==
   {0, 1, 0, 1}, {a, b, c, d}]
```

Out[14]= $\left\{\left\{a \to 0,\ b \to 0,\ c \to \frac{1}{\sqrt{2}},\ d \to \frac{1}{\sqrt{2}}\right\}\right\}$

In[15]:=
```
lösung = {a, b, c, d}.reellebasis /. sol[[1]]
```

Out[15]=
$$\frac{e^{-\frac{x}{\sqrt{2}}}\sin\left(\frac{x}{\sqrt{2}}\right)}{\sqrt{2}}+\frac{e^{\frac{x}{\sqrt{2}}}\sin\left(\frac{x}{\sqrt{2}}\right)}{\sqrt{2}}$$

Wir setzen die Lösung in die Differentialgleichung ein

In[16]:=
```
ausdruck = DE /. {y → Function[{x}, Evaluate[lösung]]}
```

Out[16]= True

In[17]:=
```
Simplify[ausdruck]
```

Out[17]= True

und berechnen die Anfangswerte:

In[18]:=
```
Table[D[lösung, {x, k}], {k, 0, 3}] /. x → 0
```

Out[18]= {0, 1, 0, 1}

Inhomogene lineare Differentialgleichungen

- **Beispiel 4.6, Übungsaufgabe 3**

In[19]:=
```
Clear[u]
```

In[20]:=
```
DE = y''[x] + 4 y'[x] + 4 y[x]
```

Out[20]= $y''(x)+4\,y'(x)+4\,y(x)$

- **homogene Gleichung**

In[21]:=
```
charpol = DE /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```

Out[21]= $\lambda^2+4\,\lambda+4$

In[22]:=
```
Solve [charpol == 0, λ]
```

Out[22]= $\{\{\lambda \to -2\},\ \{\lambda \to -2\}\}$

In[23]:=
```
DSolve[DE == 0, y[x], x]
```

Out[23]= $\left\{\left\{y(x) \to c_1\,e^{-2x}+c_2\,e^{-2x}\,x\right\}\right\}$

- **Iterative Methode: Wir lösen (nach Faktorisierung) zuerst eine inhomogene Differentialgleichung erster Ordnung:**

In[24]:=
```
lösung = DSolve[u'[x] + 2 u[x] == E^(-2 x)/x^2, u[x], x]
```

Out[24]= $\left\{\left\{u(x) \to c_1 e^{-2x} - \frac{e^{-2x}}{x}\right\}\right\}$

In[25]:=
```
U[x_] = Evaluate[u[x] /. lösung[[1]] /. C[1] → 0]
```

Out[25]= $-\frac{e^{-2x}}{x}$

- **und finden im zweiten Schritt eine spezielle Lösung der inhomogenen Differentialgleichung:**

In[26]:=
```
DSolve[y'[x] + 2 y[x] == U[x], y[x], x]
```

Out[26]= $\left\{\left\{y(x) \to c_1 e^{-2x} - e^{-2x} \log(x)\right\}\right\}$

- **Vergleich: DSolve**

In[27]:=
```
DSolve[DE == E^(-2 x)/x^2, y[x], x]
```

Out[27]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-2x} x - e^{-2x} (\log(x) + 1)\right\}\right\}$

- **Ansatzmethode**

- **Beispiel (a):**

In[28]:=
```
DE = y''''[x] - 5 y''[x] + 4 y[x]
```

Out[28]= $y^{(4)}(x) - 5\, y''(x) + 4\, y(x)$

- **homogene Gleichung**

In[29]:=
```
DSolve[DE == 0, y[x], x]
```

Out[29]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-x} + c_3 e^{x} + c_4 e^{2x}\right\}\right\}$

- **inhomogene Gleichung**

In[30]:=
```
Lösung = DSolve[DE == 1 + x + x^2, y[x], x]
```

Out[30]= $\left\{\left\{y(x) \to c_1 e^{-2x} + c_2 e^{-x} + c_3 e^{x} + c_4 e^{2x} + \frac{1}{8}\left(2x^2 + 2x + 7\right)\right\}\right\}$

- **Wir verwenden den Ansatz**

In[31]:=
```
ansatz = a x^2 + b x + c
```

Out[31]= $a x^2 + b x + c$

- **Bevor wir ihn einsetzen, schreiben wir den Ansatz als Funktion:**

In[32]:=
```
Function[{x}, Evaluate[ansatz]]
```

Out[32]= $\{x\} \quad a x^2 + b x + c$

- **und setzen dies in die inhomogene Differentialgleichung ein:**

In[33]:=
```
ausdruck = DE - (1 + x + x^2) /. {y → Function[{x}, Evaluate[ansatz]]}
```

Out[33]= $4\left(a x^2 + b x + c\right) - 10 a - x^2 - x - 1$

- **Koeffizientenvergleich liefert**

In[34]:=
```
lösung = Solve[CoefficientList[ausdruck, x] == 0, {a, b, c}]
```

Out[34]= $\left\{\left\{a \to \frac{1}{4},\ b \to \frac{1}{4},\ c \to \frac{7}{8}\right\}\right\}$

- **Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:**

In[35]:= `ansatz /. lösung[[1]]`

Out[35]= $\frac{x^2}{4}+\frac{x}{4}+\frac{7}{8}$

- **Vergleich:**

In[36]:= `Lösung`

Out[36]= $\left\{\left\{y(x)\to c_1\,e^{-2x}+c_2\,e^{-x}+c_3\,e^{x}+c_4\,e^{2x}+\frac{1}{8}\left(2x^2+2x+7\right)\right\}\right\}$

- **(b)**

In[37]:= `DE = y'''[x] - y'[x]`

Out[37]= $y^{(3)}(x)-y'(x)$

- **homogene Gleichung**

In[38]:= `DSolve[DE == 0, y[x], x]`

Out[38]= $\{\{y(x)\to c_1\,e^{x}-c_2\,e^{-x}+c_3\}\}$

- **inhomogene Gleichung**

In[39]:= `Lösung = DSolve[DE == E^x + E^(2 x), y[x], x]`

Out[39]= $\left\{\left\{y(x)\to e^{x}\left(\frac{1}{4}(4\,c_1-3)+\frac{x}{2}\right)-c_2\,e^{-x}+c_3+\frac{e^{2x}}{6}\right\}\right\}$

- **Wir verwenden den Ansatz**

In[40]:= `ansatz = a E^(2 x) + b x E^x`

Out[40]= $a\,e^{2x}+b\,e^{x}\,x$

- **und setzen dies in die inhomogene Differentialgleichung ein:**

In[41]:= `ausdruck = DE - (E^x + E^(2 x)) /. {y → Function[{x}, Evaluate[ansatz]]}`

Out[41]= $6\,a\,e^{2x}+2\,b\,e^{x}-e^{x}-e^{2x}$

- **Koeffizientenvergleich liefert**

In[42]:= `lösung = Solve[{Coefficient[ausdruck, E^x], Coefficient[ausdruck, E^(2 x)]} == 0, {a, b}]`

Out[42]= $\left\{\left\{a\to\frac{1}{6},\ b\to\frac{1}{2}\right\}\right\}$

- **Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:**

In[43]:= `ansatz /. lösung[[1]]`

Out[43]= $\frac{e^{x}\,x}{2}+\frac{e^{2x}}{6}$

- **Vergleich:**

In[44]:= `Lösung`

Out[44]= $\left\{\left\{y(x)\to e^{x}\left(\frac{1}{4}(4\,c_1-3)+\frac{x}{2}\right)-c_2\,e^{-x}+c_3+\frac{e^{2x}}{6}\right\}\right\}$

- **(c)**

In[45]:= `DE = y''[x] - 3 y'[x] + 2 y[x]`

Out[45]= $y''(x)-3\,y'(x)+2\,y(x)$

- **homogene Gleichung**

In[46]:=
```
DSolve[DE == 0, y[x], x]
```

Out[46]= $\{\{y(x)\to c_1\,e^x + c_2\,e^{2x}\}\}$

- **inhomogene Gleichung**

In[47]:=
```
Lösung = DSolve[DE == Sin[x], y[x], x]
```

Out[47]= $\left\{\left\{y(x)\to c_1\,e^x + c_2\,e^{2x} + \frac{1}{10}(\sin(x)+3\cos(x))\right\}\right\}$

- **Wir verwenden den Ansatz**

In[48]:=
```
ansatz = a Sin[x] + b Cos[x]
```

Out[48]= $a\sin(x)+b\cos(x)$

- **und setzen dies in die inhomogene Differentialgleichung ein:**

In[49]:=
```
ausdruck = DE - (Sin[x]) /. {y → Function[{x}, Evaluate[ansatz]]}
```

Out[49]= $2(a\sin(x)+b\cos(x)) - 3(a\cos(x)-b\sin(x)) - a\sin(x) - b\cos(x) - \sin(x)$

- **Koeffizientenvergleich liefert**

In[50]:=
```
lösung =
  Solve[{Coefficient[ausdruck, Sin[x]], Coefficient[ausdruck, Cos[x]]} == 0, {a, b}]
```

Out[50]= $\left\{\left\{a\to\frac{1}{10},\ b\to\frac{3}{10}\right\}\right\}$

- **Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:**

In[51]:=
```
ansatz /. lösung[[1]]
```

Out[51]= $\frac{\sin(x)}{10}+\frac{3\cos(x)}{10}$

- **Vergleich:**

In[52]:=
```
Lösung
```

Out[52]= $\left\{\left\{y(x)\to c_1\,e^x + c_2\,e^{2x} + \frac{1}{10}(\sin(x)+3\cos(x))\right\}\right\}$

- **Hausaufgabe: Ansatzverfahren für Differentialgleichung**

In[53]:=
```
DE = y'''[x] - y[x] == e^x + e^(2 x)
```

Out[53]= $y^{(3)}(x) - y(x) = e^x + e^{2x}$

- **charakteristisches Polynom**

In[54]:=
```
charpol = DE[[1]] /. {y[x] → 1, Derivative[n_][y][x] → λ^n}
```

Out[54]= $\lambda^3 - 1$

In[55]:=
```
Factor[charpol]
```

Out[55]= $(\lambda-1)(\lambda^2+\lambda+1)$

- **homogene Gleichung**

In[56]:=
```
DSolve[DE[[1]] == 0, y[x], x]
```

Out[56]= $\left\{\left\{y(x)\to c_1\,e^x + c_3\,e^{-x/2}\sin\left(\frac{\sqrt{3}\,x}{2}\right) + c_2\,e^{-x/2}\cos\left(\frac{\sqrt{3}\,x}{2}\right)\right\}\right\}$

- **Wir verwenden den Ansatz**

In[57]:=
```
ansatz = (a + b x) ⅇ^x + c ⅇ^(2 x)
```

Out[57]= $e^x (a + b x) + c\, e^{2x}$

- **Wir setzen dies in die inhomogene Differentialgleichung ein:**

In[58]:=
```
ausdruck =
  Expand[DE[[1]] - DE[[2]] /. Table[D[y[x], {x, k}] → D[ansatz, {x, k}], {k, 0, 3}]]
```

Out[58]= $3\, b\, e^x + 7\, c\, e^{2x} - e^x - e^{2x}$

- **Koeffizientenvergleich liefert**

In[59]:=
```
lösung = Solve[Join[CoefficientList[Coefficient[ausdruck, ⅇ^x], x],
    CoefficientList[Coefficient[ausdruck, ⅇ^(2 x)], x]] == 0, {a, b, c}]
```

Solve::svars : Equations may not give solutions for all "solve" variables. ≫

Out[59]= $\left\{\left\{b \to \frac{1}{3},\ c \to \frac{1}{7}\right\}\right\}$

- **Also haben wir die spezielle Lösung der inhomogenen Differentialgleichung gefunden:**

In[60]:=
```
ansatz /. lösung[[1]]
```

Out[60]= $e^x \left(a + \frac{x}{3}\right) + \frac{e^{2x}}{7}$

- **Vergleich:**

In[61]:=
```
dsol = DSolve[DE, y[x], x]
```

Out[61]= $$\left\{\left\{y(x) \to c_1\, e^x + c_3\, e^{-x/2} \sin\left(\frac{\sqrt{3}\ x}{2}\right) + c_2\, e^{-x/2} \cos\left(\frac{\sqrt{3}\ x}{2}\right) - \frac{1}{21} e^x \left(-7\, e^x - 7\, x + 4\, e^x \sin^2\left(\frac{\sqrt{3}\ x}{2}\right) + 7 \sin^2\left(\frac{\sqrt{3}\ x}{2}\right) + 4\, e^x \cos^2\left(\frac{\sqrt{3}\ x}{2}\right) + 7 \cos^2\left(\frac{\sqrt{3}\ x}{2}\right)\right)\right\}\right\}$$

In[62]:=
```
Simplify[dsol]
```

Out[62]= $$\left\{\left\{y(x) \to \frac{1}{21} e^{-x/2} \left(e^{3x/2} (21\, c_1 + 7\, x + 3\, e^x - 7) + 21\, c_3 \sin\left(\frac{\sqrt{3}\ x}{2}\right) + 21\, c_2 \cos\left(\frac{\sqrt{3}\ x}{2}\right)\right)\right\}\right\}$$

Differentialgleichungssysteme

- **Beispiel 3.1**

In[63]:=
```
dgl = {y1'[x] == -2 y1[x] + y2[x], y2'[x] == -2 y2[x] + Exp[-2 x]/x}
```

Out[63]= $\left\{\mathrm{y1}'(x) = \mathrm{y2}(x) - 2\, \mathrm{y1}(x),\ \mathrm{y2}'(x) = \frac{e^{-2x}}{x} - 2\, \mathrm{y2}(x)\right\}$

In[64]:=
```
lösung = DSolve[dgl, {y1[x], y2[x]}, x]
```

Out[64]= $\left\{\left\{\mathrm{y1}(x) \to c_2\, e^{-2x}\, x + c_1\, e^{-2x} - e^{-2x}\, x + e^{-2x}\, x \log(x),\ \mathrm{y2}(x) \to c_2\, e^{-2x} + e^{-2x} \log(x)\right\}\right\}$

schrittweise:

In[65]:=
```
lösung2 = DSolve[dgl[[2]], y2[x], x]
```

Out[65]= $\left\{\left\{\mathrm{y2}(x) \to c_1\, e^{-2x} + e^{-2x} \log(x)\right\}\right\}$

In[66]:=
```
dgl[[1]] /. lösung2[[1]]
```

Out[66]= $\mathrm{y1}'(x) = c_1\, e^{-2x} - 2\,\mathrm{y1}(x) + e^{-2x}\log(x)$

In[67]:=
```
DSolve[dgl[[1]] /. lösung2[[1]], y1[x], x]
```

Out[67]= $\{\{\mathrm{y1}(x) \to c_2\, e^{-2x} + e^{-2x}(c_1\, x - x + x\log(x))\}\}$

Vergleich:

In[68]:=
```
lösung
```

Out[68]= $\{\{\mathrm{y1}(x) \to c_2\, e^{-2x}\, x + c_1\, e^{-2x} - e^{-2x}\, x + e^{-2x}\, x\log(x),\ \mathrm{y2}(x) \to c_2\, e^{-2x} + e^{-2x}\log(x)\}\}$